

Qucs

A Report

Compact Verilog-A pn junction photodiode model

Mike Brinson

Copyright © 2008 Mike Brinson <mbrin72043@yahoo.co.uk>

Permission is granted to copy, distribute and/or modify this document under the terms of the GNU Free Documentation License, Version 1.1 or any later version published by the Free Software Foundation. A copy of the license is included in the section entitled "GNU Free Documentation License".

Introduction

Optoelectronic devices are not included in Qucs version 0.0.14 or earlier releases of the software. With the growing importance of these devices, and indeed the fact that they are present in an increasing number of electronic systems, this is a significant omission. This report presents the structure and physical details of a Qucs implementation of a pn junction photodiode model. The photodiode model is the first in a planned series of Verilog-a compact device models for optoelectronic components. The report also introduces the concept of a light bus and shows how light paths can be added to Qucs simulation schematics. A number of example schematics are also included in the report to demonstrate the performance of the new Verilog-A pn junction photodiode model. The background to the work outlined in this report was first published in the International Journal of Numerical Modelling: Electronic Networks, Devices and Fields in September 2008¹.

pn junction photodiode effects modelled

The Qucs pn junction photodiode model includes the following features:

- Diode photocurrent response characteristics expressed as a function of light power and wavelength.
- Diode DC I-V characteristics in the forward and reverse bias regions including avalanche breakdown in reverse bias.
- Diode bias dependent capacitance.
- Diode shunt resistance.
- Diode package series resistance.
- Device noise, including thermal, shot, flicker and quantum contributions.

The Qucs pn junction photodiode model

The schematic capture symbol and equivalent electrical circuit for the Qucs pn junction photodiode is shown in Fig. 1. In this model the DC properties and capacitance of the photodiode are represented by a semiconductor diode with a parallel shunt resistance R_{sh} . Device lead resistance is represented by series resistance

¹Brinson M.E. and Jahn S., Qucs: A GPL software package for circuit simulation, compact device modelling and circuit macromodelling from DC to RF and beyond, published online, 5 September 2008, <http://www3.interscience.wiley.com/journal/121397825/abstract>.

Rseries_area. A voltage-controlled current source models the diode photocurrent. The gain of the controlled source is set as the responsivity of the photodiode. The equivalent circuit shown in Fig. 1(c) presents the complete photodiode model with thermal, shot and flicker noise sources included. One interesting, and unusual, feature of the Qucs photodiode noise model is the inclusion of quantum shot noise, modelled by noise current source *Ilightn*. Photocurrent *Ilight* is given by

$$I_{light} = Light \cdot Responsivity \quad (1)$$

where *Light* is the optical signal in W and *Responsivity* is the spectral responsivity in A/W. Responsivity can also be written in terms of the photodiode quantum efficiency given by.

$$Responsivity = \frac{QE \cdot q \cdot \lambda}{h \cdot c} = \frac{QE_{percent} \cdot \lambda}{1.2398e5} \quad (2)$$

where QE is the photodiode quantum efficiency, λ is the light wavelength in nm, h is Planck's constant, c is the speed of light in a vacuum, and $QE_{percent}$ is the quantum efficiency given in percentage. In Fig. 1 the optical signal path is shown as a green line connected at the side of the photodiode symbol. In Qucs simulations optical signals are modelled as voltage quantities expressed as real numbers, even though they represent a power quantity, and are shown on a schematic as green connecting lines between components. This has the effect of clearly identifying optical signal paths on a schematic, when compared to the blue lines which indicate connecting wires characterised by conventional current and voltage signals. Similarly, signal sources which input light to a circuit are coloured green, see later example simulations.




Figure 1: Qucs pn junction photodiode model: (a) schematic capture symbol, (b) basic model circuit, (c) full equivalent circuit, including noise

Photodiode parameters

Name	Symbol	Description	Unit	Default
N	<i>N</i>	photodiode emission coefficient		1.35
Rseries	<i>Rseries</i>	series lead resistance	Ω	$1e-3$
Is	<i>Is</i>	diode dark current	A	$0.34e-12$
Bv	<i>Bv</i>	reverse breakdown voltage	V	60.0
Ibv	<i>Is</i>	current at reverse breakdown voltage	A	$1e-3$
Vj	<i>Vj</i>	junction potential	V	0.7
Cj0	<i>Cj0</i>	zero bias junction capacitance	F	$60e-12$
M	<i>M</i>	grading coefficient		0.5
Area	<i>Area</i>	diode relative area		1
Tnom	<i>Tnom</i>	parameter measurement temperature	$^{\circ}\text{C}$	26.85
Fc	<i>Fc</i>	forward-bias depletion capacitance coefficient		0.5
Tt	<i>Tt</i>	transit time	s	$10e-9$
Xti	<i>Xti</i>	saturation current temperature exponent		3.0
Eg	<i>Eg</i>	energy gap	eV	1.16
Responsivity	<i>Responsivity</i>	responsivity	A/W	0.5
Rsh	<i>Rsh</i>	shunt resistance	Ω	$5e8$
QEpercent	<i>QEpercent</i>	quantum efficiency	%	80.0
Lambda	<i>Lambda</i>	light wavelength	nm	900.0
LEVEL		responsivity calculator selector*		1
Kf	<i>Kf</i>	flicker noise coefficient		$1e-12$
Af	<i>Af</i>	flicker noise exponent		1.0
Ffe	<i>Ffe</i>	flicker noise frequency exponent		1.0
Temp	<i>Temp</i>	device temperature	$^{\circ}\text{C}$	26.85

* Parameter LEVEL is used to select how the photodiode *Responsivity* is determined: with LEVEL = 1 the model uses the listed value of parameter *Responsivity* or calculates it's value if QEpercent is not equal to zero; with LEVEL = 2 *Responsivity* is always calculated using QEpercent.

pn junction photodiode model equations

- Basic semiconductor DC characteristics

$$I_d = I_1 + I_2 + I_3 + I_4 \quad (3)$$

Where

1. $I_1 = Area \cdot Is(T2) \cdot \left[\limexp \left(\frac{V_d}{N \cdot V_t(T2)} \right) - 1 \right] + V_d \cdot GMIN \quad \forall (V_d > -5 \cdot N \cdot V_t)$
2. $I_2 = -Area \cdot Is(T2) + V_d \cdot GMIN \quad \forall (-Bv < V_d) \text{ and } (V_d > -5 \cdot N \cdot V_t)$
3. $I_3 = -I_{bv} \quad \forall (V_d = -Bv)$

$$1. Cdiff = \frac{dQdiff}{dVd} = Tt \cdot \frac{dId}{dVd}$$

$$2. Cdep = \frac{dQdep}{dVd} = Area \cdot Cj0 \cdot \left(1 - \frac{Vd}{Vj}\right)^{-M}$$

where $Qd = Qdep + Qdiff$, and

$$Qd = Tt \cdot Id + \frac{Area \cdot Cj0 \cdot Vj}{1 - M} \cdot \left\{1 - \left(1 - \frac{Vd}{Vj}\right)^{1-M}\right\} \quad \forall (Vd < Fc \cdot Vj)$$

$$Qd = Tt \cdot Id + Area \cdot Cj0 \cdot \left\{F1 + \frac{1}{F2} \cdot \left(F3 \cdot (Vd - Fc \cdot Vj) + \left[\frac{M}{2 \cdot Vj}\right] \cdot [Vd^2 - (Fc \cdot Vj)^2]\right)\right\}$$

$$\forall (Vd \geq Fc \cdot Vj)$$

Where

$$F1 = \frac{Vj}{1 - M} \cdot [1 - (1 - Fc)^{1-M}]$$

$$F2 = [1 - Fc]^{1+M}$$

$$F3 = 1 - Fc \cdot (1 + M)$$

and $GMIN = 1e-12S$.

- Diode area factors

$$1. Is_area = Is \cdot Area$$

$$2. Cjo_area = Cj0 \cdot Area$$

$$3. Rseries_area = \frac{Rseries}{Area}$$

- Diode temperature factors

$$1. Is(T2) = Is \cdot \left\{\frac{T2}{T1}\right\}^{\frac{Xti}{N}} \cdot \limexp \left\{\frac{-Eg(T1)}{Vt(T2)} \cdot \left[1 - \frac{T2}{T1}\right]\right\}$$

$$2. Vj(T2) = \frac{T2}{T1} \cdot Vj-2 \cdot Vt(T2) \cdot \ln \left(\frac{T2}{T1}\right)^{1.5} - \left\{\frac{T2}{T1} \cdot Eg(T1) - Eg(T2)\right\}$$

$$3. Cj0(T2) = Cj0 \cdot \left\{1 + M \cdot \left[400e - 6 \cdot (T2 - T1) - \frac{Vj(T2)}{Vj}\right]\right\}$$

Where

$$T1 = Tnom + 273.15, \quad T2 = Temp + 273.15$$

$$Eg(T) = EG(0) - \frac{7.02e - 4 \cdot T^2}{1108 + T}$$

$Vt = \frac{K \cdot T1}{q}$, $Vt(T2) = \frac{K \cdot T2}{q}$ and K and q have their usual meaning.

- Diode photocurrent

$$I_{light} = Light \cdot Responsivity \quad (5)$$

$$\text{Where } Responsivity = \frac{QE \cdot q \cdot \lambda}{h \cdot c} = \frac{QE_{percent} \cdot \lambda}{1.2398e55}$$

- photodiode noise

$$I_{pd_noise}^2 = IR_{shn}^2 \cdot \Delta f + Id_{sn}^2 \cdot \Delta f + Id_{fn}^2 \cdot \Delta f + I_{lightn}^2 \cdot \Delta f \quad (6)$$

$$\text{Where } IR_{shn}^2 = \frac{4 \cdot K \cdot T}{R_{sh}}, \quad Id_{sn}^2 = 2 \cdot q \cdot Id, \quad Id_{fn}^2 = \frac{Kf \cdot Id^{Af}}{f^{Ffe}},$$

$I_{light}^2 = 2 \cdot q \cdot I_{light}$, and Δf is the noise frequency bandwidth in Hz.

Verilog-A model code

```
// Qucs compact photodiode model
// The structure and theoretical background to the photodiode
// Verilog-a model are presented in the Qucs photodiode report.
//
// This is free software; you can redistribute it and/or modify
// it under the terms of the GNU General Public License as published by
// the Free Software Foundation; either version 2, or (at your option)
// any later version.
//
// Copyright (C), Mike Brinson, mbrin72043@yahoo.co.uk, October 2008.
//
#include "disciplines.vams"
#include "constants.vams"
module photodiode (Anode, Cathode, Light);
  inout Anode, Cathode, Light;
  electrical Anode, Cathode, Light;
  electrical n1;
  `define attr(txt) (*txt*)
//
  parameter real N=1.35 from [1e-6:inf]
    `attr(info="photodiode_emission_coefficient");
  parameter real Rseries=1e-3 from [1e-6:inf]
    `attr(info="series_lead_resistance" unit = "Ohm");
  parameter real Is=0.34e-12 from [1e-20:inf]
    `attr(info="diode_dark_current" unit="A" );
  parameter real Bv=60 from [1e-6:inf]
    `attr(info="reverse_breakdown_voltage" unit="V");
  parameter real Ibv=1e-3 from [1e-6:inf]
```

```

        'attr(info="current_at_reverse_breakdown_voltage" unit="A");
parameter real Vj=0.7 from [1e-6:inf]
        'attr(info="junction_potential" unit="V");
parameter real Cj0=60e-12 from [0:inf]
        'attr(info="zero-bias_junction_capacitance" unit="F");
parameter real M=0.5 from [1e-6:inf]
        'attr(info="grading_coefficient");
parameter real Area=1.0 from [1.0:inf]
        'attr(info="diode_relative_area");
parameter real Tnom=26.85 from [-273:inf]
        'attr(info="parameter_measurement_temperature" unit="Celsius");
parameter real Fc=0.5 from [1e-6:inf]
        'attr(info="forward-bias_depletion_capcitanace_coefficient");
parameter real Tt=10e-9 from [1e-20:inf]
        'attr(info="transit_time" unit="s" );
parameter real Xti=3.0 from [1e-6:inf]
        'attr(info="saturation_current_temperature_exponent");
parameter real Eg= 1.16 from [1e-6:inf]
        'attr(info="energy_gap" unit="eV");
parameter real Responsivity=0.5 from [1e-6:inf]
        'attr(info="responsivity" unit="A/W");
parameter real Rsh=5e8 from [1e-6:inf]
        'attr(info="shunt_resistance" unit="Ohm");
parameter real QEpercent=80 from [0:100]
        'attr(info="quantum_efficiency" unit="%");
parameter real Lambda=900 from [100:2000]
        'attr(info="light_wavelength" unit="nm");
parameter integer LEVEL=1 from [1:2]
        'attr(info="responsivity_calculator_selector");
parameter real Kf=1e-12 from [0:inf]
        'attr(info="flicker_noise_coefficient");
parameter real Af=1.0 from [0:inf]
        'attr(info="flicker_noise_exponent");
parameter real Ffe=1.0 from [0:inf]
        'attr(info="flicker_noise_frequency_exponent");
//
real A, B, T1, T2, F1, F2, F3, Rseries_Area, Eg_T1, Eg_T2,
real Vt_T2, Vj_T2, Cj0_T2, Is_T2, GMIN;
real I1, I2, I3, I4, I5, Id, V1, Q1, Q2, fourkt, TwoQ, Res1,
real Res2, Res, Vt, I_flicker;
real con1, con2, con3, con4, con5, con6;
// Model branches
branch (Anode, n1) b6;
branch (n1, Cathode) b1;
//
analog begin
// Model equations
@(initial_step)
begin
    Rseries_Area=(Rseries+1e-10)/Area;
    A=7.02e-4;
    B=1108.0;
    T1=Tnom+273.15;
    T2=$temperature;
    Vt='P_K*300.0/'P_Q;
    Vt_T2='P_K*T2/'P_Q;
    F1=(Vj/(1-M))*(1-pow((1-Fc),(1-M)));
    F2=pow((1-Fc), (1+M));
    F3=1-Fc*(1+M);
    Eg_T1=Eg-A*T1*T1/(B+T1);
    Eg_T2=Eg-A*T2*T2/(B+T2);
    Vj_T2=(T2/T1)*Vj-2*$vt*ln(pow((T2/T1),1.5))-((T2/T1)*Eg_T1-Eg_T2);

```

```

GMIN=1e-12;
Cj0_T2=Cj0*(1+M*(400e-6*(T2-T1)-(Vj_T2-Vj)/Vj));
Is_T2=Is*pow((T2/T1),(Xti/N))*limexp(-(Eg-T1)/$vt*(1-T2/T1));
Res1=(QEpercent != 0) ? QEpercent*Lambda/1.2398e5:Responsivity;
Res2=QEpercent*Lambda/1.2938e5;
Res=(LEVEL==1) ? Res1 : Res2;
con1=-5.0*N*Vt;
con2=Area*Is_T2;
con3=Area*Cj0_T2;
con4=Fc*Vj;
con5=Fc*Vj_T2;
con6=Bv/Vt_T2;
end;
// Current contributions
V1=V(b1);
I1=(V1 > con1) ? con2*(limexp(V1/(N*Vt_T2))-1.0)+GMIN*V1: 0;
I2=(V1 <= con1) ? -con2+GMIN*V1 :0;
I3=(V1 == -Bv)?-Ibv: 0;
I4=(V1<-Bv)?-con2*(limexp(-(Bv+V1)/Vt_T2)-1.0+con6):0;
Q1=(V1<con4) ? Tt*I1 + con3*(Vj_T2/(1-M))*(1-pow((1-V1/Vj_T2),(1-M))):0;
Q2=(V1>=con4) ? Tt*I1 + con3*(F1+(1/F2)*(F3*(V1-con5)+
(M/(2.0*Vj_T2))*(V1*V1-con5*con5))):0;
I5=V(Light)*Res;
Id=I1+I2+I3+I4;
I(b1) <+ -I5;
I(b1) <+ V(b1)/Rsh;
I(b6)<+V(b6)/Rseries_Area;
I(b1)<+Id;
I(b1)<+ddt(Q1+Q2);
I(Light)<+V(Light)/1e10;
// Noise contributions
fourkt=4.0*'P_K*$temperature;
TwoQ=2.0*'P_Q;
I_flicker=pow(Id, Af);
I(b6)<+white_noise(fourkt/Rseries_Area, "thermal");
I(b1)<+white_noise(fourkt/Rsh, "thermal");
I(b1)<+white_noise(TwoQ*Id, "shot");
I(b1)<+flicker_noise(Kf*I_flicker, Ffe, "flicker");
I(b1)<+white_noise(TwoQ*I5, "shot");
//
end
endmodule

```


Example photodiode circuits




Figure 2: Basic photodiode test circuit for simulating device output current as a function of light input level and applied DC bias




Figure 3: Photodiode test circuit for extracting device capacitance and resistance as a function of applied DC bias





Figure 4: Pulsed light power comparator circuit using two photodiodes and a logarithmic amplifier




Figure 5: Photodiode noise test circuit and simulated noise characteristics

End Note

Optoelectronic devices are an important group of electronic components and as such deserve more attention than has been given to them in the past by the Qucs development team. This situation should improve over the coming year. The pn junction photodiode model is the first in a planned series of optoelectronic models, including models for transimpedance amplifiers, optical actuators and optical media. Given time it should be possible to significantly improve Qucs optoelectronic capabilities. The photodiode model reported here is an interesting model in that it is one of the first Verilog-A models to fully utilize the recent changes to the ADMS/Qucs interface which allow proper initialisation of model parameters. A new procedure for combining Verilog-A generated models also introduces for the first time the initial stages towards a fully modularized approach to linking complex models with the main body of the Qucs code. Once again my thanks to Stefan Jahn for all his encouragement and help during the period I worked on the photodiode model.